Pensieve Header: A second attempt at the "infinitesimal Alexander module"; continues 2008-09/AlexanderEulerSpaces.nb.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of April 20, 2009, 14:18:34.482.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
(# → Alexander[#][t]) & /@ AllKnots[{3, 7}]
```

$$\{\text{Knot}[3,1]\to -1+\frac{1}{t}+t,\ \text{Knot}[4,1]\to 3-\frac{1}{t}-t,\ \text{Knot}[5,1]\to 1+\frac{1}{t^2}-\frac{1}{t}-t+t^2,$$
$$\text{Knot}[5,2]\to -3+\frac{2}{t}+2t,\ \text{Knot}[6,1]\to 5-\frac{2}{t}-2t,\ \text{Knot}[6,2]\to -3-\frac{1}{t^2}+\frac{3}{t}+3t-t^2,$$
$$\text{Knot}[6,3]\to 5+\frac{1}{t^2}-\frac{3}{t}-3t+t^2,\ \text{Knot}[7,1]\to -1+\frac{1}{t^3}-\frac{1}{t^2}+\frac{1}{t}+t-t^2+t^3,$$
$$\text{Knot}[7,2]\to -5+\frac{3}{t}+3t,\ \text{Knot}[7,3]\to 3+\frac{2}{t^2}-\frac{3}{t}-3t+2t^2,$$
$$\text{Knot}[7,4]\to -7+\frac{4}{t}+4t,\ \text{Knot}[7,5]\to 5+\frac{2}{t^2}-\frac{4}{t}-4t+2t^2,$$
$$\text{Knot}[7,6]\to -7-\frac{1}{t^2}+\frac{5}{t}+5t-t^2,\ \text{Knot}[7,7]\to 9+\frac{1}{t^2}-\frac{5}{t}-5t+t^2\}$$

```
K = Knot[4, 1];
pd = PD[K];
gc = GC @@ pd /. X[i_, j_, k_, l_] :→ If[PositiveQ[X[i, j, k, l]],
      Ar[l, i, +1], Ar[j, i, -1]
     ];
False && (gc = GC[Ar[1, 3, +1], Ar[4, 2, -1]]);
n = 2 Length[gc];
gc
```

```
GC[Ar[1, 4, 1], Ar[5, 8, 1], Ar[3, 6, -1], Ar[7, 2, -1]]
```

Conventions for red objects:

1. Legs start just to the right of the index; RedAr[0,7] means a red arrow starting to the right of position 0 (that is, to the left of everything) and ending to the right of position 7).
2. If two (red) indices are the same, the heads are to the right of the tails.
3. RedW[] is the legless wheel object.
3. RedY[i,j,k] means "red Y with tails are i and j and head at k".

```
range = Range[0, n];
AllRedObjects = Flatten[{
    Outer[RedAr, range, range], Outer[RedY, range, range, range], RedW[]
   }];
Short[AllRedObjects]
```

```
{RedAr[0, 0], RedAr[0, 1], RedAr[0, 2], RedAr[0, 3], RedAr[0, 4],
 <<801>>, RedY[8, 8, 5], RedY[8, 8, 6], RedY[8, 8, 7], RedY[8, 8, 8], RedW[]}
```

The relations associated with a red objects involve all the ways of "pulling one leg one unit to the left" :

```
RelationsIn[gc_GC, red_] := ReplaceList[
  red * (Times @@ Select[gc, (Intersection[List @@ #, List @@ red] =!= {}) &]), {
   (* Short red arrows are central: *)
   (* RedAr[i_, i_]*_. /; i> 0 :> -red+RedAr[i-1, i-1], *)
   (* Tails commute for RedAr: *)
   RedAr[i_, j_] Ar[i_, _, _] * _. :>
    -red + RedAr[i - 1, j] + If[i - 1 == j, -x RedW[], 0],
   (* Commuting a RedAr tail across an Ar head *)
   RedAr[i_, j_] Ar[k_, i_, s_] * _. :>
    -red + RedAr[i - 1, j] + (X^s - 1) / x RedY[k, i, j] + If[i - 1 == j, -x RedW[], 0],
   (* Commuting the head of a RedAr with the tail of an Ar *)
   RedAr[i_, j_] Ar[j_, k_, s_] * _. :>
    -red + RedAr[i, j - 1] + (X^s - 1) / x RedY[j, i, k - 1] + If[i == j, x RedW[], 0],
   (* Commuting the head of a RedAr with the head of an Ar *)
   RedAr[i_, k_] Ar[j_, k_, s_] * _. :>
    -red + RedAr[i, k - 1] - (X^s - 1) / x RedY[j, i, k - 1] + If[i == k, x RedW[], 0],
   (* The anti-symmetry of RedY: *)
   RedY[i_, j_, k_] * _. :> red + RedY[j, i, k],
   (* Tails commute for RedY: *)
   RedY[i_, j_, k_] Ar[i_, _, _] * _. :>
    -red + RedY[i - 1, j, k] + If[i - 1 == k, x^2 RedW[], 0],
   RedY[i_, j_, k_] Ar[j_, _, _] * _. :>
    -red + RedY[i, j - 1, k] + If[j - 1 == k, -x^2 RedW[], 0],
   (* Commuting a RedY tail across an Ar head *)
   RedY[i_, j_, k_] Ar[l_, j_, s_] * _. :>
    -red + RedY[i, j - 1, k] - (X^s - 1) RedY[l, j, k] + If[j - 1 == k, -x^2 RedW[], 0],
   (* Commuting a RedY head across an Ar tail *)
   RedY[i_, j_, k_] Ar[k_, l_, s_] * _. :>
    -red + RedY[i, j, k - 1] - (X^s - 1) RedY[i, j, l - 1] - (
      If[i == k, x^2 RedW[], 0] + If[j == k, -x^2 RedW[], 0]
     ),
   (* Commuting a RedY head across an Ar head *)
   RedY[i_, j_, l_] Ar[k_, l_, s_] * _. :>
    -red + RedY[i, j, l - 1] + (X^s - 1) RedY[i, j, l - 1] - (
      If[i == l, x^2 RedW[], 0] + If[j == l, -x^2 RedW[], 0]
     )
  }
 ]
```

```
AllRedRelations = Flatten[RelationsIn[gc, #] & /@ AllRedObjects];
rule = Thread[Rule[AllRedObjects, IdentityMatrix[Length[AllRedObjects]]]];
Short[RedRules = Map[
   (
     p = First@Part[AllRedObjects, First@Position[#, 1, {1}]];
     p → p - (#.AllRedObjects)
    ) &,
   DeleteCases[mat = Simplify[RowReduce[AllRedRelations /. rule]], {0 ...}]
  ]]
```

{RedAr[0, 0] → RedAr[8, 8], RedAr[0, 1] → RedAr[8, 8], ≪805≫, RedY[8, 8, 7] → 0, RedY[8, 8, 8] → 0}

```
Simplify[RedRules[[Table[Random[Integer, {1, Length[RedRules]}], {10}]]]]
```

$\left\{\text{RedY}[7,6,1]\to 0,\ \text{RedY}[2,5,1]\to -\frac{X^2\,(-1+X)\,\text{RedW}[]}{1-3X+X^2},\right.$
$\text{RedY}[3,5,5]\to \frac{X^2\,(-1+X)^2\,\text{RedW}[]}{1-3X+X^2},\ \text{RedY}[0,2,8]\to 0,$
$\text{RedY}[2,1,2]\to -X^2\left(1+\frac{1}{1-3X+X^2}\right)\text{RedW}[],\ \text{RedY}[4,5,7]\to 0,$
$\text{RedY}[6,8,5]\to -\frac{X^2\left(1-3X+2X^2\right)\text{RedW}[]}{X\left(1-3X+X^2\right)},\ \text{RedY}[8,6,3]\to \frac{X^2\,(-1+X)^2\,\text{RedW}[]}{X\left(1-3X+X^2\right)},$
$\left.\text{RedY}[8,6,1]\to -\frac{X^2\,(-1+X)^3\,\text{RedW}[]}{X\left(1-3X+X^2\right)},\ \text{RedY}[0,8,5]\to -\frac{X^2\,\text{RedW}[]}{X}\right\}$

```
RedZ = Plus @@ gc /. Ar[i_, j_, s_] :→ s * RedAr[i, j]
```

RedAr[1, 4] - RedAr[3, 6] + RedAr[5, 8] - RedAr[7, 2]

```
Simplify[RedZ /. RedRules]
```

$$\frac{X\,(2-3X)\,\text{RedW}[]}{1-3X+X^2}$$